

5.2 Klassische Beispiele der Mikrophysik

Wir kennen nun den Zusammenhang von Welle und Teilchen: Die Wirkungsfunktion und die Interferenz bestimmen den Ablauf des Geschehens. D.h., im Falle eines einzelnen Teilchens läßt sich die Wirkungswelle fast unmittelbar angeben, sobald die Erhaltungsgrößen und das Potential bekannt sind. Wir müssen keine Bewegungsgleichung mehr lösen, weil wir mit den Lösungen selbst arbeiten können. Versuchen wir es mit den Bewegungen im homogenen Feld und im Zentralfeld.

5.2.1 Der Wurf

Wir untersuchen die Wirkungswellen der klassischen Wurfbewegung so, wie man es in der Schulphysik auch macht: Zuerst kommt der senkrechte Wurf und dann durch Überlagerung mit einer gleichförmigen Bewegung der schiefe Wurf. Der Unterschied zur klassischen Behandlung besteht nur darin, daß wir mit Wellen arbeiten, die interferieren können.

`wiwurf.ms`

Senkrechter Wurf

Wir benötigen den Impuls, die kinetische Energie und die Lagrangefunktion als Funktionen der Gesamtenergie H und der Höhe y .

Energie-Impuls-Beziehung:

$$> \text{Ep} := p^2 / (2 * m) = T;$$

Energieerhaltung:

$$> \text{Ees} := H = T + V;$$

Lagrange-Funktion:

$$> \text{Lagr} := L = T - V;$$

potentielle Energie

$$> V := m * g * y;$$

$$E_p := \frac{1}{2} \frac{p^2}{m} = T$$

$$E_{es} := H = T + V$$

$$Lagr := L = T - V$$

$$V := m g y$$

$$> \text{sol} := \text{solve}(\{\text{Ep}, \text{Ees}, \text{Lagr}\}, \{p, T, L\});$$

$$\text{sol} := \{T = H - m g y, L = H - 2 m g y,$$

$$p = \text{RootOf}(-Z^2 - 2 m H + 2 m^2 g y)\}$$

```
> allvalues(");
```

$$\left\{ T = H - m g y, L = H - 2 m g y, p = \sqrt{2 m H - 2 m^2 g y} \right\}, \left\{ T = H - m g y, L = H - 2 m g y, p = -\sqrt{2 m H - 2 m^2 g y} \right\}$$

```
> assign(sol);
> allvalues(p);
```

$$\sqrt{2 m H - 2 m^2 g y}, -\sqrt{2 m H - 2 m^2 g y}$$

```
> py:=op(1,["]);
```

$$p y := \sqrt{2 m H - 2 m^2 g y}$$

Damit erhalten wir für den senkrechten Wurf einer Masse m mit der Gesamtenergie H bei der Fallbeschleunigung g die charakteristische Funktion w

```
> w:=int(py,y);
```

$$w := -\frac{1}{3} \frac{(2 m H - 2 m^2 g y)^{3/2}}{m^2 g}$$

und als ortsabhängigen Teil der Wirkungswelle und ihres Realteils:

```
> psi:=exp(I*w);
> pr:=Re(psi);
```

$$\psi := e^{\left(-1/3 \frac{I(2 m H - 2 m^2 g y)^{3/2}}{m^2 g} \right)}$$

Dabei wurde zunächst die vereinfachende Annahme gemacht, daß die Amplitude der Wirkungswelle 1 ist. Wir können nun schon den Realteil von ψ darstellen (Abb. 5.1 links):

```
> m:=1/2:g:=10:H:=50: ys:=H/(m*g);
> plr:=plot(evalc(pr),y=0..15,numpoints=1000):
> #stackfehler fuer numpoints=2000
> plr;
```

$$y_s := 10$$

Zunächst erkennt man den Zusammenhang „großer Impuls – kleine Wellenlänge“. Dann sollte bei der klassischen Steighöhe $y_s = H/(mg)$ die Wellenlänge unendlich groß werden, also die Amplitude konstant bleiben? Statt dessen geht sie auf Null zurück! Wie läßt sich das erklären? Oberhalb y_s wird der Impuls imaginär, und das hat ein exponentielles Abklingen der Amplitude zur Folge, denn es gilt $\psi \sim \exp(ip_y y)$. Andererseits verschwindet die Wirkungswelle auch nicht schlagartig, wie man es von einem klassischen Ansatz her erwarten würde. Dieses Phänomen ist als eine typisch quantenmechanische Erscheinung




Abb. 5.1: Wirkungswelle zum senkrechten Wurf

bekannt: der Tunneleffekt! Im Gegensatz zum Standard-Tunneleffekt an einem rechteckigen Potentialwall endlicher Höhe und Breite kann unser Teilchen aber nicht *durch*tunneln, weil das Potential ohne Grenzen linear ansteigt. Wir haben also mit dem einfachen Ansatz der Wirkungswelle Ψ auf Anhieb ein Ergebnis bekommen, das man sonst erst nach mühseligen Erörterungen zur Lösung der Schrödingergleichung erhält. Im Worksheet können Sie nun wieder die Parameter ändern, den Imaginärteil von Ψ darstellen (der oberhalb von y verschwinden muß) und 3D-Darstellungen erzeugen (Abb. 5.1 rechts).

```
> plot3d(evalc(pr), x=0..1, y=0..15, axes=boxed, grid=[10,100],
> orientation=[-40,10]);
```

Diese ebenen Wellen bedeuten, daß die x -Koordinate des geworfenen Teilchens nicht festliegt. Allein das Potential und die Startgeschwindigkeit (also die gesamte Energie) bestimmen den Vorgang.

Interferenz der aufsteigenden mit der absteigenden Welle.

Wenn die Wirkungswellen echte Wellen sind, muß sich wegen der Reflexion im Scheitelpunkt eine stehende Welle ergeben, allerdings mit ortsabhängiger Wellenlänge. Im Gegensatz zu den gewohnten Wellen können wir aber die Laufrichtung nicht durch das Vorzeichen von Ht umkehren, denn es gilt immer $S = W - Ht$ (und nie $S = W + Ht$). Die Laufrichtung wird vielmehr durch den Impulsvektor bestimmt, der im eindimensionalen Fall das Vorzeichen von W ändert. Das ist unterhalb des Scheitels unproblematisch. Aber oberhalb des Scheitels wird W imaginär positiv, wie Sie dem Plot im Worksheet entnehmen können.

```
> w;
```

$$-\frac{2}{15}(50-5y)^{3/2}$$

```
> m:=1/2:g:=10:H:=50:
> plot({w,Im(w)},y=-yS..2*yS);
```

Wir dürfen also oberhalb von y_s das Vorzeichen von W nicht ändern, sonst bekommen wir eine unphysikalische (exponentiell ansteigende) Lösung.

```
> m:='m':g:='g':H:='H':t:='t':yS:='yS':
> Sp:=w-H*t;
```

Herausfiltern des unphysikalischen Teils mit `conjugate`

```
> Sm:=-conjugate(w)-H*t;
```

$$Sp := -\frac{1}{3} \frac{(2mH - 2m^2gy)^{3/2}}{m^2g} - Ht$$
$$Sm := \frac{1}{3} \text{conjugate} \left(\frac{(2mH - 2m^2gy)^{3/2}}{m^2g} \right) - Ht$$

Für die Interferenz müssen wir die Amplituden der Wirkungswellen von Sp und Sm addieren:

```
> psiint:=exp(I*Sp)+exp(I*Sm);
```

$$\text{psiint} := e^{i \left(-\frac{1}{3} \frac{(2mH - 2m^2gy)^{3/2}}{m^2g} - Ht \right)}$$
$$+ e^{i \left(\frac{1}{3} \text{conjugate} \left(\frac{(2mH - 2m^2gy)^{3/2}}{m^2g} \right) - Ht \right)}$$

Realteil der stehenden Welle

```
> t:='t':
> rpsiint:=evalc(Re(psiint));
```

$$\text{rpsiint} := e^{\left(\frac{1}{3} \frac{|\%1|^{3/2} \sin(\%2)}{m^2g} \right)} \cos \left(\frac{1}{3} \frac{|\%1|^{3/2} \cos(\%2)}{m^2g} + Ht \right)$$
$$+ e^{\left(\frac{1}{3} \frac{|\%1|^{3/2} \sin(\%2)}{m^2g} \right)} \cos \left(-\frac{1}{3} \frac{|\%1|^{3/2} \cos(\%2)}{m^2g} + Ht \right)$$
$$\%1 := 2mH - 2m^2gy$$
$$\%2 := \left(\frac{3}{4} - \frac{3}{4} \text{signum}(\%1) \right) \pi$$

```
> m:=1/2:g:=10:H:=50:t:=0.01:
> plot(rpsiint,y=-5..12,a=-2..2,numpoints=1000);
```

Eine Momentaufnahme steht natürlich. Aber steht auch die Welle?

```
> t:='t':
> animate(rpsiint,y=0..12,t=0..2*Pi/H,numpoints=500);
```

Lokalisierung

Realistischer als die unendlich ausgedehnte Welle ist ein halbwegs lokalisiertes Teilchen, das den Umkehrpunkt zur Zeit $t = 0$ erreicht. Wir brauchen zur Simulation der Reflexion zwei davon: Reflexion = Überlagerung zweier gegenläufiger Pakete (jenseits des Umkehrpunktes werden die Pakete durch die Amplitude der Wirkungswelle automatisch ausgeblendet s.u.). Zunächst die Mittelpunkte zweier spiegelbildlich zu y_s laufender Pakete

```
> y01:=ys+1/2*g*t^2;  
> y02:=ys-1/2*g*t^2;
```

$$y01 := ys + \frac{1}{2} g t^2$$

$$y02 := ys - \frac{1}{2} g t^2$$

dann die Gaußfunktionen

```
> paket1:=exp(-(y-y01)^2/s^2);  
> paket2:=exp(-(y-y02)^2/s^2);  
> total:=paket1+paket2;
```

$$paket1 := e\left(-\frac{(y-y_s-1/2 g t^2)^2}{s^2}\right)$$

$$paket2 := e\left(-\frac{(y-y_s+1/2 g t^2)^2}{s^2}\right)$$

$$total := e\left(-\frac{(y-y_s-1/2 g t^2)^2}{s^2}\right) + e\left(-\frac{(y-y_s+1/2 g t^2)^2}{s^2}\right)$$

```
> ys:=H/m/g;
```

$$ys := \frac{H}{m g}$$

```
> m:=1/2:g:=10:H:=50:s:=2:  
> total;
```

$$e\left(-1/4(y-10-5 t^2)^2\right) + e\left(-1/4(y-10+5 t^2)^2\right)$$

Wir können testen:

```
> animate(total,y=-ys..2*ys,t=-2..2);
```

In diese Einhüllenden setzen wir die Wirkungswellen von oben

```
> m:='m':g:='g':H:='H':s:='s':  
> Sp:=w-H*t;  
> Sm:=-conjugate(w)-H*t;
```

$$Sp := -\frac{1}{3} \frac{(2 m H - 2 m^2 g y)^{3/2}}{m^2 g} - H t$$

$$S_m := \frac{1}{3} \operatorname{conjugate} \left(\frac{(2 m H - 2 m^2 g y)^{3/2}}{m^2 g} \right) - H t$$

Interferenz (nach Multiplikation mit den Paketfunktionen)

```
> psiint:=exp(I*Sp)*paket1+exp(I*Sm)*paket2;
```

$$\begin{aligned} \text{psiint} := & e \left(I \left(-1/3 \frac{(2 m H - 2 m^2 g y)^{3/2}}{m^2 g} - H t \right) \right) e \left(-\frac{(y - \frac{H}{m g} - 1/2 g t^2)^2}{s^2} \right) \\ & + e \left(I \left(1/3 \operatorname{conjugate} \left(\frac{(2 m H - 2 m^2 g y)^{3/2}}{m^2 g} \right) - H t \right) \right) e \left(-\frac{(y - \frac{H}{m g} + 1/2 g t^2)^2}{s^2} \right) \end{aligned}$$

```
> rpsiint:=evalc(Re(psiint));
```

$$\begin{aligned} \text{rpsiint} := & e \left(1/3 \frac{|\%1|^{3/2} \sin(\%2)}{m^2 g} \right) \cos \left(\frac{1}{3} \frac{|\%1|^{3/2} \cos(\%2)}{m^2 g} + H t \right) \\ & e \left(-\frac{(y - \frac{H}{m g} - 1/2 g t^2)^2}{s^2} \right) + e \left(1/3 \frac{|\%1|^{3/2} \sin(\%2)}{m^2 g} \right) \\ & \cos \left(-\frac{1}{3} \frac{|\%1|^{3/2} \cos(\%2)}{m^2 g} + H t \right) e \left(-\frac{(y - \frac{H}{m g} + 1/2 g t^2)^2}{s^2} \right) \\ \%1 := & 2 m H - 2 m^2 g y \\ \%2 := & \left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1) \right) \pi \end{aligned}$$

Bitte warten...

```
> m:=1/2:g:=10:H:=50:s:=3:
> animate(rpsiint,y=-ys..2*ys,t=-2..2,numpoints=400,
> frames=25);
```

Die Sache klappt also im Prinzip für den senkrechten Wurf, d.h., wir haben soeben mit geringem Aufwand die WKB-Näherung „für das im Schwerfeld tanzende Elektron“ (vgl. [22]) erzeugt und sichtbar gemacht!

Schiefer Wurf

Für den schiefen Wurf müssen wir in x-Richtung eine Bewegung mit konstantem Impuls p_x überlagern:

```
> m:='m':g:='g': H:='H':px:='px':V:='V':
> ys:=(H-1/2*px^2/m)/m/g;
```

$$y_s := \frac{H - \frac{1}{2} \frac{p_x^2}{m}}{m g}$$

> py:=sqrt(2*m*(H-V)-px^2);

$$py := \sqrt{2mH - 2mV - px^2}$$

> V:=m*g*y;

$$V := mgy$$

> w:=int(py,y)+px*x;

$$w := -\frac{1}{3} \frac{(2mH - 2m^2gy - px^2)^{3/2}}{m^2g} + px^2$$

Zur Kontrolle die Bahn, die man aus der Wirkungsfunktion am einfachsten so erhält:

> ypl:=diff(w,px);

$$ypl := \frac{\sqrt{2mH - 2m^2gy - px^2}}{m^2g} + x$$

> ypl:=solve(ypl,y);

$$ypl := \frac{1}{2} \frac{2mH - px^2 - \frac{x^2 m^4 g^2}{px^2}}{m^2g}$$

> m:=1/2:g:=10:H:=50:px:=2:

> plot(ypl,x=-8..8);

Nachdem wir also mit der richtigen Wirkungsfunktion rechnen, können wir wieder zur Wellenfunktion zurückkehren

> m:='m':g:='g':H:='H':px:='px':

> rpsiS:=evalc(Re(exp(I*(w-H*t))));

$$rpsiS := e^{\left(\frac{1}{3} \frac{|\%1|^{3/2} \sin\left(\left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1)\right) \pi\right)}{m^2g}\right)} \cos\left(\frac{1}{3} \frac{|\%1|^{3/2} \cos\left(\left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1)\right) \pi\right)}{m^2g} - px^2 + Ht\right)$$

$$\%1 := 2mH - 2m^2gy - px^2$$

Im Gegensatz zum senkrechten Wurf (ebene Wellenfronten) müssen jetzt die Wellenfronten gekrümmt sein, denn sie sind die Orthogonaltrajektorien zu einer Parabelschar, die durch Parallelverschiebung in x-Richtung entsteht. Diese Neilschen Parabeln können Sie im Worksheet *wiwurf1.ms* mit Methoden der Differentialgeometrie erzeugen und untersuchen (Abb. 5.2 links). Es gibt aber eine einfachere Methode: Wir stellen die Wellen selbst dar (Abb. 5.2 rechts).

```

> px:=6:H:=50:g:=10:m:=1/2:n:=8:
> animate3d(rpsiS,x=0..2,y=0..1.5*ys,t=0..2*Pi/H*(1-1/n),
> axes=boxed,grid=[50,20],orientation=[-40,20],
> frames=n,style=wireframe);

```




Abb. 5.2: Links: Die eingezeichneten Neilschen Parabeln stehen senkrecht auf den aufsteigenden Ästen der Wurfparabeln und sind somit die Wellenfronten zu positivem p_y . Rechts: Die entsprechenden Wirkungswellen, die Sie im Worksheet laufen lassen können.

Wenn Sie im Worksheet zurückgehen und das Vorzeichen von p_y ändern, bekommen Sie die Wellen zur Abwärtsbewegung. Für die gesamte Bewegung muß also der Typ der Wellen im Umkehrpunkt geändert werden (s.u.).

Man darf sich allerdings beim Experimentieren mit diesen Darstellungen vom Bild nicht täuschen lassen: wenn Sie die Parameter ändern, werden Sie feststellen, daß die Wellen unerwartete Muster aufweisen können. Das liegt an der Interpolation, die von den Maple-Plotroutinen gemacht wird, also auch an der Orientierung von 3D-Plots. In solchen Fällen muß man die Auflösung erhöhen oder ändern (grid) bzw. einen Contourplot erstellen (Abb. 5.3).

```

> t:=0:px:=6:
> contourplot(rpsiS,x=0..2,y=0..1.5*ys,axes=boxed,
> grid=[60,50],contours=5);

```

Schiefer Wurf mit Interferenz

Ein altes Problem des Übergangs von der klassischen Physik zu Quantenphysik, mit dem sich schon Schrödinger und Sommerfeld geplagt haben, ist: Wie sind die Randbedingungen bzw. Anfangsbedingungen zu wählen? In der Quantenmechanik = Wellenmechanik gibt es keine wirkliche Bahn mehr. Es gibt „nur“




Abb. 5.3: Wellenfronten zum schiefen Wurf, vgl. Abb. 5.2

noch Erhaltungsgrößen. Ein Zustand wird durch Interferenz gebildet. Aber Interferenz wovon? Fällt das Teilchen gerade oder steigt es? (Es macht zu jeder Zeit beides.) Von wo aus steigt es oder fällt es? Welcher Ast der (mindestens zweiwertigen) Funktion der Wellenfronten ist zu nehmen? Diese Fragen führen im Falle periodischer Bewegungen zu den Wirkungs- und Winkelvariablen. In unserem Beispiel des schiefen Wurfes (aperiodischer Grenzfall = Parabel) können wir diesen Überlegungen mit der Unterstützung von Maple in der Art eines Designers nachgehen, der seine Modelle Schritt für Schritt entwickelt. Dazu lassen wir wieder wie beim senkrechten Wurf zwei Pakete gegen den Umkehrpunkt laufen (und darüber hinaus):

```
> y01:=ys+1/2*g*t^2*signum(t);
> y02:=ys-1/2*g*t^2*signum(t);
```

$$y01 := ys + \frac{1}{2} g t^2 \text{signum}(t)$$

$$y02 := ys - \frac{1}{2} g t^2 \text{signum}(t)$$

```
> ys:=(H-1/2*px^2/m)/m/g;
```

$$ys := \frac{H - \frac{1}{2} \frac{px^2}{m}}{m g}$$

```
> paket1:=exp(-((y-y01)^2+(x-px/m*t)^2)/s^2);
> paket2:=exp(-((y-y02)^2+(x-px/m*t)^2)/s^2);
```

$$paket1 := \exp \left(- \frac{\left(y - \frac{H - 1/2 \frac{px^2}{m}}{mg} - 1/2 g t^2 \operatorname{signum}(t) \right)^2 + \left(x - \frac{px t}{m} \right)^2}{s^2} \right)$$

Für die Plots ist es günstig, wenn wir die Steigzeit und die Wurfweite zur Verfügung haben:

```
> ts:=sqrt(2*ys/g);
> xw:=abs(px)/m*ts;
```

$$ts := \sqrt{2} \sqrt{\frac{H - \frac{1}{2} \frac{px^2}{m}}{m g^2}}$$

$$xw := \frac{|px| \sqrt{2} \sqrt{\frac{H - \frac{1}{2} \frac{px^2}{m}}{m g^2}}}{m}$$

Nun noch die charakteristischen Funktionen für die Aufwärts- und Abwärtsbewegung

```
> wp:=w;
> wm:=conjugate(-w)+2*px*x;
```

$$wp := -\frac{1}{3} \frac{(2 m H - 2 m^2 g y - p x^2)^{3/2}}{m^2 g} + p x x$$

$$wm := \operatorname{conjugate} \left(\frac{1}{3} \frac{(2 m H - 2 m^2 g y - p x^2)^{3/2}}{m^2 g} - p x x \right) + 2 p x x$$

und die Amplituden der Wirkungswellen

```
> rpsip:=evalc(Re(exp(I*wp)));
```

$$rpsip := \exp \left(\frac{1}{3} \frac{|\%1|^{3/2} \sin \left(\left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1) \right) \pi \right)}{m^2 g} \right)$$

$$\cos \left(-\frac{1}{3} \frac{|\%1|^{3/2} \cos \left(\left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1) \right) \pi \right)}{m^2 g} + p x x \right)$$

$$\%1 := 2 m H - 2 m^2 g y - p x^2$$

```
> rpsim:=evalc(Re(exp(I*wm)));
```

$$rpsim := \exp \left(\frac{1}{3} \frac{|\%1|^{3/2} \sin \left(\left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1) \right) \pi \right)}{m^2 g} \right)$$

$$\cos \left(\frac{1}{3} \frac{|\%1|^{3/2} \cos \left(\left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1) \right) \pi \right)}{m^2 g} + p x x \right)$$

$$\%1 := 2 m H - 2 m^2 g y - p x^2$$

Als zeitabhängige Wirkungsfunktionen erhalten wir

- > Sp:=wp-H*t;
- > Sm:=wm-H*t;

$$Sp := -\frac{1}{3} \frac{(2 m H - 2 m^2 g y - p x^2)^{3/2}}{m^2 g} + p x x - H t$$

$$Sm := \operatorname{conjugate} \left(\frac{1}{3} \frac{(2 m H - 2 m^2 g y - p x^2)^{3/2}}{m^2 g} - p x x \right) + 2 p x x - H t$$

Nun können wir die Wirkungswellen zum auf- und absteigenden Ast zur Interferenz bringen

- > wurf:=paket1*exp(I*Sp)+paket2*exp(I*Sm);

$$\begin{aligned} \text{wurf} := & e \left(-\frac{\left(y - \frac{H-1/2 \frac{p x^2}{m}}{m g} - 1/2 g t^2 \operatorname{signum}(t) \right)^2 + \left(x - \frac{p x t}{m} \right)^2}{s^2} \right) \\ & e \left(I \left(-1/3 \frac{(2 m H - 2 m^2 g y - p x^2)^{3/2}}{m^2 g} + p x x - H t \right) \right) + \\ & e \left(-\frac{\left(y - \frac{H-1/2 \frac{p x^2}{m}}{m g} + 1/2 g t^2 \operatorname{signum}(t) \right)^2 + \left(x - \frac{p x t}{m} \right)^2}{s^2} \right) \\ & e \left(I \left(\operatorname{conjugate} \left(1/3 \frac{(2 m H - 2 m^2 g y - p x^2)^{3/2}}{m^2 g} - p x x \right) + 2 p x x - H t \right) \right) \end{aligned}$$

Realteil

- > Rwurf:=evalc(Re(wurf));

$$\begin{aligned} R\text{wurf} := & e \left(-\frac{\left(y - \frac{H}{m g} + 1/2 \frac{p x^2}{m^2 g} - 1/2 g t^2 \operatorname{signum}(t) \right)^2 - \left(x - \frac{p x t}{m} \right)^2}{s^2} \right) \\ & e \left(1/3 \frac{|\%1|^{3/2} \sin(\%2)}{m^2 g} \right) \cos \left(\frac{1}{3} \frac{|\%1|^{3/2} \cos(\%2)}{m^2 g} - p x x + H t \right) \end{aligned}$$

$$\begin{aligned}
& + e \left(\frac{-\left(y - \frac{H}{m g} + 1/2 \frac{p_x^2}{m^2 g} + 1/2 g t^2 \operatorname{signum}(t)\right)^2 - \left(x - \frac{p_x t}{m}\right)^2}{s^2} \right) \\
& e \left(1/3 \frac{|\%1|^{3/2} \sin(\%2)}{m^2 g} \right) \cos \left(-\frac{1}{3} \frac{|\%1|^{3/2} \cos(\%2)}{m^2 g} - p_x x + H t \right) \\
\%1 & := 2 m H - 2 m^2 g y - p_x^2 \\
\%2 & := \left(\frac{3}{4} - \frac{3}{4} \operatorname{signum}(\%1) \right) \pi
\end{aligned}$$

Zahlenwerte

```

> m:=1/2:g:=10:H:=50:
> px:=2:
> t:=-1/2*ts:s:=2:

```

Momentaufnahmen von oben bzw. von schräg oben zeigen, daß sich eine Animation lohnt:

```

> plot3d(Rwurf,x=-xw..xw,y=0..2*ys,axes=boxed,
> grid=[50,50],orientation=[-90,0],
> style=contour,contours=5);
> plot3d(Rwurf,x=-xw..xw,y=0..2*ys,axes=boxed,
> grid=[50,20],orientation=[-40,2],
> style=wireframe);

```

Parameter für die Animation

```

> n:=18:t:='t':s:=2:px:=1:

```

Bitte warten... oder kleineres n wählen:

```

> animate3d(10^10*Rwurf,x=-xw-s..xw+s,y=0..1.5*ys,
> t=-ts..ts,axes=boxed,orientation=[-90,0],
> grid=[50,50],frames=n,st

```

Abbildung 5.4 zeigt drei Momentaufnahmen des schiefen Wurfes eines Mikroteilchens. Im Worksheet können Sie mit einem 486er die Animation ohne weiteres mit 18 Bildern laufen lassen. Stellen Sie auf Vollbild, und genießen Sie das Schauspiel! Ein paar Gleichungen, ein CAS und die handelsübliche Hardware (von gestern) liefern uns ein Mikroskop in die Welt der elementaren Vorgänge, von dem wohl schon Heisenberg geträumt hat. Wenn man bedenkt, daß wir hier die Tür zu einer neuen Dimension im Umgang mit der Physik öffnen, so nimmt man gerne ein paar apparative Probleme in Kauf, das ist das Los der Pioniere: leider erzeugt Maple durch Interpolationsfehler bei linearer Darstellung ein Maschenmuster. Logarithmieren bzw. Quadrieren der Funktion beheben das, beseitigen aber auch die interessanten Wellenfronten $S = 0$, auf denen das Teilchen zu reiten scheint – mit stroboskopischen Effekten. Das Quadrat (oder eine gerade Hochzahl) hat einen unerwünschten Nebeneffekt im Scheitel, weil




Abb. 5.4: Schiefer Wurf, quantenmechanisch. Sogar Heisenberg würde hier sein Mikroskop scharf stellen... bis zu einem gewissen Quantum läßt Maple das auch zu.

Maximum und Minimum dann gleich bewertet werden. Man kann aber mit 10^{10} multiplizieren (dann aber nicht 1:1 wählen), um die Interpolationskreuze zum Verschwinden zu bringen.

Für PostScript-Fans, die ihre .ps-files gerne am laufenden Band produzieren möchten, hier die zugehörigen Befehle:

```
t:='t':vxlab='':
for t from -1 to 1 do
vtitle:='t=' .t: 'Name':='p.(6+t).wiwurf.' .'.ps:pspl(Name):
plot3d(10^10*Rwurf,x=-xw-s..xw+s,y=0..1.5*ys,axes=framed,
orientation=[-90,0],grid=[100,100],style=contour,
shading=z,contours=5,opt3d);winpl():
od;
```

Zur Ergänzung noch die 3D-Animation

```
> n:=15:t:='t':
> animate3d(Rwurf,x=-xw-s..xw+s,y=0..1.5*ys,
> t=-ts..ts,axes=boxed,frames=n,grid=[20,20],
> orientation=[-40,55]);
```

Wir treiben die *Simulation* noch einen Schritt weiter und lassen das Teilchen entstehen und vergehen:

```
> s:=1+abs(t);
```

$$s := 1 + |t|$$

Bitte warten, bis das Teilchen entstanden ist...

```
> animate3d(10^10*Rwurf,x=-xw-2..xw+2,y=0..1.5*ys,
> t=-ts..ts,axes=boxed,orientation=[-90,0],
> grid=[50,50],frames=n);
```

Während das Elektron im Scheitel eine Schrödingersche Zitterbewegung macht, taucht unten Diracs See auf... verkehrte Welt?

Ein letzter Handgriff, dann sollte das Design stimmen: Amplitude³ umgekehrt proportional zur Zeit

```
> n:=25;  
> animate3d(Rwurf/s,x=-xw-2..xw+2,y=0..1.5*ys,  
> t=-ts..ts,axes=boxed,frames=n,grid=[20,20],  
> orientation=[-40,55]);
```

wiwurf.ms

Damit haben wir den Wurf quantenmechanisch behandelt, ohne komplizierte Formeln lösen zu müssen, denn mit dem Ansatz der Wirkungswellen und ihrer Interferenz konnten wir schon von Lösungen ausgehen, die wir nur der Situation anpassen mußten. Wir können uns nun *vorstellen*, wie sich ein Elektron oder Atom in einem homogenen Feld bewegt. Das ist nicht nur von akademischem Interesse, es kann auch im Realexperiment beobachtet werden! Die Schwierigkeit bei diesen Experimenten liegt aber nicht darin, ein Teilchen möglichst hoch zu werfen (wie etwa bei Sportwettkämpfen), vielmehr hat man Mühe, es so weit abzukühlen, daß seine Wellenlänge groß genug wird, um die Interferenz in der Nähe des Umkehrpunktes beobachten zu können. Die Kühlung von Atomen erreicht man, indem man sie in eine Falle von Laserstrahlen einschließt. Kennen Sie ein entsprechendes Experiment mit Elektronen? Aber spielen Sie doch noch ein bißchen mit den Computerteilchen – ohne jedes apparative Problem.

³„Korrekt“ wäre Amplitudenquadrat, vgl. Feynman-Propagator.